STANDARD ML

LISTS

a list is an **immutable** finite sequence of elements

```
[3, 5, 9]: int list
["a", "list"]: str list
[]: 'a list
```

order matters

```
[1, 2, 3] <> [3, 2, 1];
val it = true : bool
```

and repetitions count

```
[3, 3, 3] <> [3];
val it = true : bool
```

elements may have any type

```
[(1,"One"),(2,"Two")] : (int*string) list
[[3.1],[],[5.7, ~0.6]]: real list list
```

... but all elements must have the same type

```
[5, "five"]; (*ERROR*)
```

```
stdIn:3.1-3.12 Error: operator and operand don't agree [overload conflict]
  operator domain: [int ty] * [int ty] list
  operand:         [int ty] * string list
  in expression:
    5 :: "five" :: nil
```

the empty list has a polymorphic type

```
[]: 'a list
```

`nil` is a synonym of `[]`

```
nil;
val it = [] : 'a list
```

BUILDING A LIST

a list is either *empty* or *a head followed by a tail*

`[1,2,3]` ➪ head: `1` tail: `[2,3]`

use the infix operator `::` (aka `cons`) to build a list

```
1 :: [2, 3];

1 :: 2 :: 3 :: [];
```

```
val it = [1,2,3] : int list
val it = [1,2,3] : int list
```

`::` associates to the right, so

```
x1 :: x2 :: ... :: xn :: nil
```

=

```
(x1 :: (x2 :: (... :: (xn :: nil)...)
```

`::` is a *constructor* so it can be used in patterns

```
fun replace_head (_::t) x = x :: t
  | replace_head [] _ = []
;
```

```
val replace_head = fn : 'a list -> 'a -> 'a list
```

BUILTIN FUNDAMENTAL FUNCTIONS

`null` - tests whether a list is empty

```
fun null [] = true
  | null (_::_) = false;
```

```
val null = fn : 'a list -> bool
```

`hd` - evaluates to the head of a non-empty list

```
fun hd (x::_) = x;
```

```
stdIn:7.5-7.18 Warning: match nonexhaustive
          x :: _ => ...

val hd = fn : 'a list -> 'a
```

```
hd[ [ [1,2], [3] ], [ [4] ] ];
```
```
val it = [[1,2],[3]] : int list list
```

```
hd it;
```
```
val it = [1,2] : int list
```

```
hd it;
```
```
val it = 1 : int
```

`tl` - evaluates to the tail of a non-empty list

```
fun tl (_::xs) = xs;
```

```
stdIn:11.5-11.20 Warning: match nonexhaustive
          _ :: xs => ...

val tl = fn : 'a list -> 'a list
```

```
tl ["how", "are", "you?"];
val it = ["are","you?"] : string list
```

```
tl it;
val it = ["you?"] : string list
```

```
tl it;
val it = [] : string list
```

```
tl it;

uncaught exception Match [nonexhaustive match failure]
  raised at: stdIn:11.20
```

EXAMPLE - BUILDING A LIST OF INTEGERS

```
fun range (m, n) =
  if m = n then []
  else m :: (range (m+1, n));

range (2, 5);
```

```
val range = fn : int * int -> int list
val it = [2,3,4] : int list
```

```
infix --;
val op-- = range;

2 -- 5;
```

```
infix --
val -- = fn : int * int -> int list
val it = [2,3,4] : int list
```

`take` AND `drop`

$$xs = [x_1, x_2, x_3, \ldots, x_k, x_{k+1}, \ldots, x_n]$$

$$take(k, xs) = [x_1, x_2, x_3, \ldots, x_k]$$

$$drop(k, xs) = [x_{k+1}, \ldots, x_n]$$

THE COMPUTATION OF `take`

```
fun take (0, _)     = []
  | take (i, x::xs) = x :: (take (i-1, xs));
```

```
stdIn:17.5-17.69 Warning: match nonexhaustive
          (0,_) => ...
          (i,x :: xs) => ...

val take = fn : int * 'a list -> 'a list
```

```
take (3, [9,8,7,6,5,4])
9 :: take (2, [8,7,6,5,4])
9 :: (8 :: take (1, [7,6,5,4]))
9 :: (8 :: (7 :: take (0, [6,5,4])))
9 :: (8 :: (7 :: []))
9 :: (8 :: [7])
9 :: [8,7]
[9,8,7]
```

THE COMPUTATION OF `drop`

```
fun drop (0, xs)    = xs
  | drop (i, _::xs) = drop (i-1, xs);
```

```
stdIn:18.5-18.62 Warning: match nonexhaustive
          (0,xs) => ...
          (i,_ :: xs) => ...

val drop = fn : int * 'a list -> 'a list
```

```
drop (3, [9,8,7,6,5,4])
drop (2,   [8,7,6,5,4])
drop (1,     [7,6,5,4])
drop (0,       [6,5,4])
[6,5,4]
```

TAIL RECURSION

normal recursion

```
fun take(0, _)      = []
  | take(i, x::xs) = x::(take(i-1, xs));
```

tail recursion

```
fun drop (0, xs)     = xs
  | drop (i, _::xs) = drop (i-1, xs);
```

NORMAL TO TAIL RECURSIVE

```
fun length []      = 0
  | length (_::xs) = 1 + length xs;
```

```
val length = fn : 'a list -> int
```

use an **accumulator** to make it iterative

```
local
  fun ilen (n, [])    = n
    | ilen (n, _::xs) = ilen (n+1, xs)
in
  fun length xs = ilen (0, xs)
end;
```

```
val length = fn : 'a list -> int
```

BUILTIN APPEND OPERATOR

`[x1,...,xm] @ [y1,...,yn] =`
`[x1,...,xm,y1,...,yn]`

```
infix @;
fun []      @ ys = ys
  | (x::xs) @ ys = x :: (xs @ ys);

["Append", "is"] @ ["never", "boring"];
```

```
infix @
val @ = fn : 'a list * 'a list -> 'a list
val it = ["Append","is","never","boring"] : string list
```

- is it tail recursive?
- why can't it be used in patterns?

SIDE NOTE - `orelse` AND `andalso`

they are short-circuiting boolean operators

```
B1 andalso B2 = if B1 then B2  else false;

B1 orelse  B2 = if B1 then true else B2;
```

```
fun even n = (n mod 2 = 0);

fun powoftwo n =
  (n=1) orelse
  (even n andalso powoftwo (n div 2));
```

```
val even = fn : int -> bool
val powoftwo = fn : int -> bool
```

is `powoftwo` tail-recursive?

BUILTIN FUNCTION `map`

```
fun map f []      = []
  | map f (x::xs) = (f x) :: (map f xs);
```

```
val map = fn : ('a -> 'b) -> 'a list -> 'b list
```

```
val sqlist = map (fn x => x*x);

sqlist [1,2,3];
```

```
val sqlist = fn : int list -> int list
val it = [1,4,9] : int list
```

transposing a matrix using `map`


```
fun transp ([]::_) = []
  | transp rows =
      (map hd rows) :: (transp (map tl rows));
```

```
val transp = fn : 'a list list -> 'a list list
```

BUILTIN FUNCTION `filter`

```
fun filter pred []      = []
  | filter pred (x::xs) =
      if pred x then (x:: filter pred xs)
                else      filter pred xs;
```

```
val filter = fn : ('a -> bool) -> 'a list -> 'a list
```

```
filter (fn x => x mod 2 = 0) [1,2,3,4,5];
```

```
val it = [2,4] : int list
```

`filter` is bound as `List.filter`

USING `map` AND `filter`

a polynomial is represented as a list of $(coeff, degree)$ pairs

$$5x^3 + 2x + 7$$

```
type polynomial = (int*int) list;
val a = [(5,3), (2,1), (7,0)]: polynomial;
```

```
type polynomial = (int * int) list
val a = [(5,3),(2,1),(7,0)] : polynomial
```

taking the derivative of a polynomial

```
fun derive (p: polynomial): polynomial =
    List.filter
        (fn (coeff, deg) => deg >= 0)
        (map
            (fn (coeff, deg) => (coeff*deg, deg-1))
            p
        )
;

derive a;
```

```
val derive = fn : polynomial -> polynomial
val it = [(15,2),(2,0)] : polynomial
```

`find`

```
fun find f [] = NONE
  | find f (x::xs) = if f x then SOME x else find f xs;
```

```
val find = fn : ('a -> bool) -> 'a list -> 'a option
```

bound as `List.find`

`foldl` AND `foldr`

BUILTIN FUNCTION `foldl`

```
fun foldl f init []      = init
  | foldl f init (x::xs) = foldl f (f (x, init)) xs;
```

```
val foldl = fn : ('a * 'b -> 'b) -> 'b -> 'a list -> 'b
```

calculates

$$[x_1, x_2, \ldots, x_n] \longrightarrow f(x_n, \ldots, f(x_2, f(x_1, init)))$$

BUILTIN FUNCTION `foldr`

```
fun foldr f init []      = init
  | foldr f init (x::xs) = f (x, foldr f init xs);
```

```
val foldr = fn : ('a * 'b -> 'b) -> 'b -> 'a list -> 'b
```

calculates

$$[x_1, x_2, \ldots, x_n] \rightarrow f(x1, \ldots, f(xn-1, f(xn, init)))$$

USING `foldl` AND `foldr`

let's redefine some functions...

```
fun sum l = foldl op+ 0 l;

fun reverse l = foldl op:: [] l;

fun xs @ ys = foldr op:: ys xs;
```

```
val sum = fn : int list -> int
val reverse = fn : 'a list -> 'a list
val @ = fn : 'a list * 'a list -> 'a list
```

`exists` **AND** `all`

BUILTIN FUNCTION `exists`

```
fun exists p []      = false
  | exists p (x::xs) = (p x) orelse exists p xs;
```

```
val exists = fn : ('a -> bool) -> 'a list -> bool
```

checks if the predicate `p` is satisfied by at least one element of the list

```
exists (fn x => x < 0) [1, 2, ~3, 4];
```

```
val it = true : bool
```

bound as `List.exists`

BUILTIN FUNCTION `all`

```
fun all p []      = true
  | all p (x::xs) = (p x) andalso all p xs;
```

```
val all = fn : ('a -> bool) -> 'a list -> bool
```

checks if the predicate `p` is satisfied by **all** elements of the list

```
all (fn x => x >= 0) [1, 2, ~3, 4];
```

```
val it = false : bool
```

bound as `List.all`

```
fun disjoint (xs, ys) =
  all (fn x => all (fn y => x<>y) ys) xs;
```

```
stdIn:39.54-39.56 Warning: calling polyEqual
val disjoint = fn : ''a list * ''a list -> bool
```

EQUALITY IN POLYMORPHIC FUNCTIONS

equality is polymorphic in a restricted sense

- defined for values constructed of integers, strings, booleans, chars, tuples, lists and datatypes
- not defined for values containing
  - functions: equality is undecidable (halting problem)
  - reals, because e.g. nan != nan
  - elements of abstract types

ML has a polymorphic equality type `''a`

```
op=;
```

```
stdIn:40.1-40.4 Warning: calling polyEqual
val it = fn : ''a * ''a -> bool
```

somewhat like an interface/trait in other languages

EXAM QUESTIONS

EXERCISE 1

implement `map` using `foldl`

```
val foldl = fn : ('a * 'b -> 'b) -> 'b -> 'a list -> 'b;
val map = fn : ('a -> 'b) -> 'a list -> 'b list;
```

```
fun map f inpList = foldl
    _
    _
    inpList
;
```

```
map (fn x => x * 2) [1,2,3,4];
```

EXERCISE 2

`insSort` (insertion sort) sorts a list according to a given less-then function.

```
val foldr = fn : ('a * 'b -> 'b) -> 'b -> 'a list -> 'b;
val insSort : ('a * 'a -> bool) -> 'a list -> 'a list;
```

```
fun insSort lt inpList = foldr
    _
    _
    inpList
;
```

```
insSort (op<) [1, ~3, 5, 0];
```

EXERCISE 3

```
fun upto m n = if (m > n)
    then []
    else m::(upto (m+1) n)
;

infix o;
fun f o g = fn x => f (g x);
```

```
val upto = fn : int -> int -> int list
infix o
val o = fn : ('a -> 'b) * ('c -> 'a) -> 'c -> 'b
```

what will be printed?

```
val a = map (upto 2) (upto 2 5);
```

```
val a = [[2],[2,3],[2,3,4],[2,3,4,5]] : int list list
```

what will be printed?

```
map
    (
        (fn f => null (f()))
        o
        (fn t => fn () => tl t)
    )
    a
;
```

```
val it = [true,false,false,false] : bool list
```

what will be printed?

```
map
    (List.filter (fn t => t mod 2 = 0))
    a
;
```

```
val it = [[2],[2],[2,4],[2,4]] : int list list
```

EXERCISE 4

implement a tail recursive `append`

reminder:

```
infix @;
fun []        @ ys = ys
  | (x::xs) @ ys = x :: (xs @ ys);
```

```
fun append ...
```

EXERCISE 5

implement `flatten` using `foldr`

```
flatten : 'a list list -> 'a list;
```

```
fun flatten ...
```

```
[1,2,3,4,5,6,7,8,9] = flatten [[1,2,3],[4,5,6],[],[7,8,9]];
```